```
> de:=diff(M(t), t)=r*M(t)-P;
  r:=0.06;
```

$$de := \frac{d}{dt} M(t) = 0.06\, M(t) - P$$

$$r := 0.06 \tag{1}$$

```
> sol:=dsolve( {de, M(0)=500000}, M(t));
```

$$sol := M(t) = \frac{50\,P}{3} + e^{\frac{3t}{50}} \left(500000 - \frac{50\,P}{3} \right) \tag{2}$$

```
> eq:=eval(rhs(sol), t=30.0) =0;
```

$$eq := -84.16079103\,P + 3.024823732 \times 10^{6} = 0 \tag{3}$$

```
> annualpayment := solve(eq, P);
```

$$annualpayment := 35941.00881 \tag{4}$$

```
> r:=0.06/365;
  P:=35941.00881/12;
```

$$r := 0.0001643835616$$

$$P := 2995.084068 \tag{5}$$

```
> dayspermonth:=[31,28,31,30,31,30,31,31,30,31,30,31];
```

$$dayspermonth := [31, 28, 31, 30, 31, 30, 31, 31, 30, 31, 30, 31] \tag{6}$$

```
> M:=500000;
   for y to 30 do
     for m in dayspermonth do
       for d to m do M:= M + r*M; od:
        M:=M-P;
  od;
  od:
  M;
```

$$M := 500000$$

$$7005.001342 \tag{7}$$

```
> M:=500000;
   for y to 30 do
    for m in dayspermonth do
     for d to m do
       M:=M + r*M;
        if d=15 then M:= M-P; fi;
  od:
  od;
  od:
  M;
```

$$M := 500000$$

$$-648.9168967 \tag{8}$$

```
>
> with(plots):

  M := 500000:
  r := 0.06/365:
  P := 35941.00881/12:
  dayspermonth := [31,28,31,30,31,30,31,31,30,31,30,31]:

  pts := [[0,M]]:
```

```
t := 0:

for y to 30 do
  for m in dayspermonth do
    for d to m do
      M := M + r*M;
      if d=15 then
        M := M - P;
      end if;
      t := t+1;
      pts := [op(pts), [t/365., M]];
    end do;
  end do;
end do:

listplot(pts, labels=["Years","Balance"], title="Loan Balance Decay");
```




```
>
```